Earthquake Response Spectra—ERES

The program ERES (**Ea**rthquake **R**esponse **S**pectra) is a subroutine subprogram that calculates the absolute acceleration response spectrum, the relative velocity response spectrum, and the relative displacement response spectrum of an input acceleration time history for specified damping factors. At the same time, it also calculates the maximum acceleration, velocity, and displacement of the input time history. Which spectrum is to be calculated is specified by the argument *IND2*.

ERES（**Ea**rthquake **R**esponse **S**pectra）

【Purpose】

To calculate the absolute acceleration response spectrum, relative velocity response spectrum, or relative displacement response spectrum of the input acceleration time history for specified damping factors. At the same time, calculate the maximum acceleration, maximum velocity, or maximum displacement of the input time history.

【Usage】

(1) How to connect

```
CALL ERES (NH, H, ND1, NT, T, ND2, DT, NN, DDY, ND3, IND, QMAX, RES)
```

Argument	Type	Parameter in calling program	Return Parameter
NH	I	Total number of damping factors	Unchanged
H	R 1-D array (ND1)	Damping factors (non-dimension)	Unchanged
ND1	I	Dimension size of H and RES in calling program (ND1 .GE. NH)	Unchanged
NT	I	Total number of periods to calculate response	Unchanged
T	R 1-D array (ND2)	Periods to calculate response (unit : sec)	Unchanged
ND2	I	Dimension size of T and RES in calling program (ND2 .GE. NT)	Unchanged
DT	R	Time increment of acceleration time history (unit: sec)	Unchanged

NN	I	Total number of acceleration time history data	Unchanged
DDY	R 1-D array (ND3)	Given input acceleration time history (unit : Gal)	Unchanged
ND3	I	Dimension size of DDY in calling program (ND3 .GE. NN)	Unchanged
IND	I	Index for calculation 1 : Absolute acceleration spectrum 2 : Relative velocity spectrum 3 : Relative displacement spectrum	Unchanged
QMAX	R	No need to input here	If IND=1, maximum value of input acceleration (unit : Gal) If IND=2, maximum value of input velocity (unit : cm/sec) If IND=3, maximum value of input displacement (unit: cm)
RES	R 2-D array (ND2,ND1)	No need to input here	If IND=1, absolute acceleration response spectrum (unit : Gal) If IND=2, relative velocity response spectrum (unit : cm/sec) If IND=3, relative displacement response spectrum (unit: cm)

(2) Necessary subroutines and function subprograms

None

【Calculation Method】

The response time history of a single-mass-damping system with an undamped eigen period T and a damping constant h, subjected to a ground motion acceleration $\ddot{y}(t)$, can be obtained by solving the following equation of motion.

$$\ddot{x} + 2h\varpi\dot{x} + \varpi^2 x = -\ddot{y}$$

where $\varpi = 2\pi/T$.

If the calculated acceleration, velocity, and displacement response time histories are $\ddot{x}(t)$, $\dot{x}(t)$, and $x(t)$, respectively, then the absolute acceleration, relative velocity, and displacement response spectra are expressed by $S_a(T,h) = |\ddot{x} + \ddot{y}|_{\max}$, $S_v(T,h) = |\dot{x}|_{\max}$, $S_d(T,h) = |x|_{\max}$, respectively.

In the first half of this program, 'MAXIMA OF INP UT MOTION' block calculates the input acceleration and the maximum values of velocity and displacement integrated from it. The calculation method is the same as that of the program **IACC**, but while **IACC** calculates the time histories of the velocity and displacement, the time histories are not necessary here, so only the maximum values are calculated.

In the second half of the program, 'RESPONSE COMPUTATION' block performs the same calculations as in the separate subroutine RESP, which calculates the seismic response of a single mass damping system for each given damping constant and period, but again the response time history is not required, so only the maximum response time history is calculated. Acceleration response, velocity response, or displacement response is specified by the argument *IND*.

The results are contained in the two-dimensional array *RES*, where the first index of the two-dimensional array *RES* corresponds to the period and the second index to the damping constant. That is, for example, if *IND* = 1, then *RES* (17, 2) contains the absolute acceleration response values for the 17th period in array *T* and the second damping factor in array *H*.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * *
C     SUBROUTINE FOR EARTHQUAKE RESPONSE SPECTRA
C  * * * * * * * * * * * * * * * * * * * * * * * * *
C
C                              CODED BY Y.OHSAKI
C
C     PURPOSE
C       TO COMPUTE ABSOLUTE ACCELERATION, RELATIVE VELOCITY OR
C       RELATIVE DISPLACEMENT RESPONSE SPECTRA OF THE GIVEN ACCELERA-
C       TION TIME HISTORY FOR SPECIFIED DAMPING FACTORS. THE MAXIMUM
C       INPUT ACCELERATION, VELOCITY OR DISPLACEMENT ARE ALSO COMPUTED
C
C     USAGE
C       CALL ERES(NH,H,ND1,NT,T,ND2,DT,NN,DDY,ND3,IND,QMAX,RES)
C
C     DESCRIPTION OF ARGUMENTS
C       NH             - TOTAL NUMBER OF DAMPING FACTORS
C       H(ND1)         - DAMPING FACTORS IN DECIMAL FRACTION
C       ND1            - DIMENSION OF H,RES IN CALLING PROGRAM  ND1.GE.NH
C       NT             - TOTAL NUMBER OF PERIODS FOR RESPONSE COMPUTATION
C       T(ND2)         - PERIODS IN SEC FOR RESPONSE COMPUTATION
C       ND2            - DIMENSION OF T,RES IN CALLING PROGRAM  ND2.GE.NT
C       DT             - TIME INCREMENT IN THE ACCELERATION TIME HISTORY
C                        IN SEC
C       NN             - TOTAL NUMBER OF DATA IN THE ACCELERATION TIME
C                        HISTORY
C       DDY(ND3)       - THE GIVEN ACCELERATION TIME HISTORY IN GALS
C       ND3            - DIMENSION OF DDY IN CALLING PROGRAM  ND3.GE.NN
C       IND            - 1 : FOR ABSOLUTE ACCELERATION RESPONSE SPECTRA
C                        2 : FOR RELATIVE VELOCITY RESPONSE SPECTRA
C                        3 : FOR RELATIVE DISPLACEMENT RESPONSE SPECTRA
C       QMAX           - IF IND=1 : MAX.  INPUT ACCELERATION
C                        IF IND=2 : MAX.  INPUT VELOCITY
C                        IF IND=3 : MAX.  INPUT DISPLACEMENT
C       RES(ND2,ND1)   - RESPONSE SPECTRA CORRESPONDING TO THE IND
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C       NONE
C
      SUBROUTINE ERES(NH,H,ND1,NT,T,ND2,DT,NN,DDY,ND3,IND,QMAX,RES)
C
      DIMENSION H(ND1),T(ND2),DDY(ND3),RES(ND2,ND1)
```

```
      DIMENSION EMAX(3),RMAX(3)
      PARAMETER (P2=6.283185)
C
C     MAXIMA OF INPUT MOTION
C
      EMAX(1)=ABS(DDY(1))
      EMAX(2)=0.
      EMAX(3)=0.
      DDYF=DDY(1)
      DYF=0.
      YF=0.
      DO 110 M=2,NN
      DDYM=DDY(M)
      DY=DYF+(DDYF+DDYM)*DT/2.
      Y=YF+DYF*DT+(DDYF/3.+DDYM/6.)*DT**2
      EMAX(1)=AMAX1(EMAX(1),ABS(DDYM))
      EMAX(2)=AMAX1(EMAX(2),ABS(DY))
      EMAX(3)=AMAX1(EMAX(3),ABS(Y))
      DDYF=DDYM
      DYF=DY
      YF=Y
  110 CONTINUE
      QMAX=EMAX(IND)
C
C     RESPONSE COMPUTATION
C
      DO 150 L=1,NH
      DO 140 K=1,NT
      IF(T(K).EQ.0.) GO TO 130
      W=P2/T(K)
      W2=W*W
      HW=H(L)*W
      WD=W*SQRT(1.-H(L)**2)
      WDT=WD*DT
      E=EXP(-HW*DT)
      CWDT=COS(WDT)
      SWDT=SIN(WDT)
      A11= E*(CWDT+HW*SWDT/WD)
      A12= E*SWDT/WD
      A21=-E*W2*SWDT/WD
      A22= E*(CWDT-HW*SWDT/WD)
      SS=-HW*SWDT-WD*CWDT
      CC=-HW*CWDT+WD*SWDT
      S1=(E*SS+WD)/W2
      C1=(E*CC+HW)/W2
      S2=(E*DT*SS+HW*S1+WD*C1)/W2
      C2=(E*DT*CC+HW*C1-WD*S1)/W2
      S3=DT*S1-S2
      C3=DT*C1-C2
      B11=-S2/WDT
      B12=-S3/WDT
      B21=(HW*S2-WD*C2)/WDT
      B22=(HW*S3-WD*C3)/WDT
```

```
      RMAX(1)=2.*HW*ABS(DDY(1))*DT
      RMAX(2)=ABS(DDY(1))*DT
      RMAX(3)=0.
      DXF=-DDY(1)*DT
      XF=0.
      DO 120 M=2,NN
      DDYM=DDY(M)
      DDYF=DDY(M-1)
      X= A12*DXF+A11*XF+B12*DDYM+B11*DDYF
      DX=A22*DXF+A21*XF+B22*DDYM+B21*DDYF
      DDX=-2.*HW*DX-W2*X
      RMAX(1)=AMAX1(RMAX(1),ABS(DDX))
      RMAX(2)=AMAX1(RMAX(2),ABS(DX))
      RMAX(3)=AMAX1(RMAX(3),ABS(X))
      DXF=DX
      XF=X
  120 CONTINUE
      RES(K,L)=RMAX(IND)
      GO TO 140
  130 RES(K,L)=0.
      IF(IND.EQ.1) RES(K,L)=EMAX(1)
  140 CONTINUE
  150 CONTINUE
      RETURN
      END
```

【Example】

Read the time history of the seismic motion from the file EQ.01 and obtain the acceleration response spectra for damping factors h = 0, 5, and 10%. The response is calculated for the period T given to the DATA statement. If the *IND* of the DATA statement is 2 or 3, the velocity response spectrum and displacement response spectrum can be obtained, respectively.

```
C
      DIMENSION H(3),T(35),DDY(800),RES (35,3)
      DATA      NH/3/,  H/0.0,0.05,0.10/,IND/1/
      DATA      NT/35/,T/0.00,0.05,0.10,0.15,0.20,0.25,0.30,0.35,0.40,
     *                   0.45,0.50,0.55,0.60,0.65,0.70,0.75,0.80,0.85,
     *                   0.90,0.95,1.00,1.20,1.40,1.60,1.80,2.00,2.20,
     *                   2.40,2.60,2.80,3.00,3.50,4.00,4.50,5.00/
C
      READ(5,501) DT,NN,(DDY(M),M=1,NN)
C
      CALL ERES (NH,H,3,NT,T,35,DT,NN,DDY,800,IND,AMX,RES)
      STOP
  501 FORMAT(T51,F10.0,I10/(8F10.0))
      END
```

Output：The acceleration response spectra for damping factors h = 0, 5, and 10% are stored in *RES* (T ,1), *RES* (T ,2), and *RES* (T ,3), and can be plotted as shown below.